<commit_before>set _tempvbs=%LocalAppData%\getadmin.vbs

:: Check for admin permissions
Net Session >nul 2>&1
:: If error flag set, we do not have admin.
If ERRORLEVEL 1 (
   Echo Requesting administrative privileges...
   Goto sub_elevate
) Else ( Goto sub_main )

:sub_elevate

   echo Set objShell = CreateObject^("Shell.Application"^) > %_tempvbs%
   echo objShell.ShellExecute "%~f0", "", "", "runas", 1 >> %_tempvbs%
   cscript "%_tempvbs%" //nologo
   Exit /B

:sub_main

   if exist "%_tempvbs%" ( Del "%_tempvbs%" )

   pushd %USERPROFILE%\dot-files

   if not exist ".vim\bundle\Vundle.vim" ( git clone <EMAIL>:gmarik/Vundle.vim.git .vim\bundle\Vundle.vim )

   if not exist "%USERPROFILE%\.vim" ( mklink /D %USERPROFILE%\.vim %USERPROFILE%\dot-files\.vim )

   if not exist "%USERPROFILE%\.vimrc" (mklink %USERPROFILE%\.vimrc %USERPROFILE%\dot-files\.vimrc )
   if not exist "%USERPROFILE%\.bashrc" (mklink %USERPROFILE%\.bashrc %USERPROFILE%\dot-files\.bashrc)
   if not exist "%USERPROFILE%\.bash_profile" (mklink %USERPROFILE%\.bash_profile %USERPROFILE%\dot-files\.bash_profile)
   if not exist "%USERPROFILE%\.azure-completion.sh" (mklink %USERPROFILE%\.azure-completion.sh %USERPROFILE%\dot-files\.azure-completion.sh)

   pause

   vim +PluginInstall +qall

   pause<commit_msg>Change the install script to use relative directory.<commit_after>set _tempvbs=%LocalAppData%\getadmin.vbs

setlocal
@echo off

:: Check for admin permissions
Net Session >nul 2>&1
:: If error flag set, we do not have admin.
If ERRORLEVEL 1 (
   Echo Requesting administrative privileges...
   Goto sub_elevate
) Else ( Goto sub_main )

:sub_elevate
   echo Set objShell = CreateObject^("Shell.Application"^) > %_tempvbs%
   echo objShell.ShellExecute "%~f0", "", "", "runas", 1 >> %_tempvbs%
   cscript "%_tempvbs%" //nologo
   Exit /B

:sub_main
   if exist "%_tempvbs%" ( Del "%_tempvbs%" )

   :: Script should be in the root of the dot-files directory - wherever that is.
   set fileRoot=%~dp0

   pushd %fileRoot%

   if not exist ".vim\bundle\Vundle.vim" (
      echo "Downloading Vundle (vim package manager) into .vim\bundle\Vundle.vim"
      git clone <EMAIL>:gmarik/Vundle.vim.git .vim\bundle\Vundle.vim
   )

   if not exist ".vim\swap\" (
      mkdir .vim\swap
   )

   if not exist "%USERPROFILE%\.vim" (mklink /D %USERPROFILE%\.vim %fileRoot%.vim)
   if not exist "%USERPROFILE%\.vimrc" (mklink %USERPROFILE%\.vimrc %fileRoot%.vimrc)
   if not exist "%USERPROFILE%\.bashrc" (mklink %USERPROFILE%\.bashrc %fileRoot%.bashrc)
   if not exist "%USERPROFILE%\.bash_profile" (mklink %USERPROFILE%\.bash_profile %fileRoot%.bash_profile)

   echo Continue to install Vundle plugins
   pause
   :: Install Vundle Plugins
   vim +PluginInstall +qall

   echo Install completed
   pause
   popd

endlocal
{'fraction_non_alphanumeric': 0.13516405135520684, 'fraction_numerical': 0.0039229671897289585, 'mean_word_length': 3.770408163265306, 'pattern_counts': {'":': 0, '<': 5, '<?xml version=': 0, '>': 15, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 1, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1001358', 'n_tokens_mistral': 1073, 'n_tokens_neox': 955, 'n_words': 255}